HYPERBOLA

In the projectiviced affine euclidean space and with respecto the orthonormal coordinate system R={O;B={e_1,e_2}} we consider the projective conic C of equation $x_1^2 - x_2^2 + 2x_0x_2 - 2x_0x_1 - x_0^2 = 0$.

```
> restart; with(linalg):with(plots):
> A:=matrix(3,3,[-1,-1,1,-1,1,0,1,0,-1]);
```

$$A := \begin{bmatrix} -1 & -1 & 1 \\ -1 & 1 & 0 \\ 1 & 0 & -1 \end{bmatrix} \quad (1)$$

Classify.

```
> A00:=submatrix(A,2..3,2..3);
```

$$A00 := \begin{bmatrix} 1 & 0 \\ 0 & -1 \end{bmatrix} \quad (1.1)$$

```
> det(A00); det(A);
```

$$-1$$
$$1 \quad (1.2)$$

HYPERBOLA.

Obtain the center, the axes and the asymptotes.

Proper center, pole of the line at infinity.

```
> Center:=evalm(inverse(A)&*matrix(3,1,[1,0,0]));
```

$$Center := \begin{bmatrix} -1 \\ -1 \\ -1 \end{bmatrix} \quad (2.1)$$

The asymptotes are the polar lines of the points at infinity of the hyperbola. The hyperbola is the only conic with real asymptotes. The go through the center.

```
> X:=matrix(3,1,[x[0],x[1],x[2]]);
```

(2.2)

$$X := \begin{bmatrix} x_0 \\ x_1 \\ x_2 \end{bmatrix} \quad (2.2)$$

```
> conic:=evalm(transpose(X)&*A&*X)[1,1];
```

$$conic := \left(-x_0 - x_1 + x_2\right) x_0 + \left(-x_0 + x_1\right) x_1 + \left(x_0 - x_2\right) x_2 \quad (2.3)$$

```
> solve({conic,x[0]=0});
```

$$\{x_0 = 0, x_1 = x_2, x_2 = x_2\}, \{x_0 = 0, x_1 = -x_2, x_2 = x_2\} \quad (2.4)$$

```
> infi1:=matrix(3,1,[0,-1,1]);infi2:=matrix(3,1,[0,1,1]);
```

$$infi1 := \begin{bmatrix} 0 \\ -1 \\ 1 \end{bmatrix}$$

$$infi2 := \begin{bmatrix} 0 \\ 1 \\ 1 \end{bmatrix} \quad (2.5)$$

```
> asin1:=evalm(transpose(infi1)&*A&*X)[1,1]=0;asin2:=evalm
(transpose(infi2)&*A&*X)[1,1]=0;
```

$$asin1 := 2\, x_0 - x_1 - x_2 = 0$$

$$asin2 := x_1 - x_2 = 0 \quad (2.6)$$

The axes of the conics with proper center are the lines going throught the center and the improper points given by the eigenvectors associated to the distinct eigenvalues of A00.

```
> eigenvectors(A00);
```

$$\left[-1, 1, \left\{\left[\begin{array}{cc} 0 & 1 \end{array}\right]\right\}\right], \left[1, 1, \left\{\left[\begin{array}{cc} 1 & 0 \end{array}\right]\right\}\right] \quad (2.7)$$

```
> v1:=matrix(3,1,[0,1,0]);v2:=matrix(3,1,[0,0,1]);
```

$$v1 := \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}$$

$$v2 := \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \quad (2.8)$$

```
> axis1:=det(concat(Center,v1,X))=0;axis2:=det(concat(Center,
v2,X))=0;
```

$$axis1 := x_0 - x_2 = 0$$

$$axis2 := -x_0 + x_1 = 0 \quad (2.9)$$

PLOTTING

```
> Xafin:=matrix(3,1,[1,x,y]):Cafin:=simplify(evalm(transpose
  (Xafin)&*A&*Xafin))[1,1];GconicaAfin:=implicitplot({Cafin},x=
  -4..4,y=-4..4,color=red,axes=normal):
```

$$Cafin := -1 - 2\,x + 2\,y + x^2 - y^2 \tag{2.10}$$

```
> Geje1Afin:=implicitplot({1-y},x=-4..4,y=-4..4,color=green,
  axes=normal):
> Geje2Afin:=implicitplot({-1+x},x=-4..4,y=-4..4,color=green,
  axes=normal):
> Gasintota1Afin:=implicitplot({x-y},x=-4..4,y=-4..4,color=
  blue,axes=normal):
> Gasintota2Afin:=implicitplot({-y-x+2},x=-4..4,y=-4..4,color=
  blue,axes=normal):
> display([GconicaAfin,Geje1Afin,Geje2Afin, Gasintota1Afin,
  Gasintota2Afin]);
```



Invariants, reduced equation in terms of the invariants and coodinate system in which the equation of the conic is the reduced equation.

Invariants:

```
> det(A);det(A00);eigenvalues(A00);
```

$$1$$

$$-1$$

$$1, -1 \quad (3.1)$$

```
> lambda[1]:=1; lambda[2]:=-1;
```

$$\lambda_1 := 1$$

$$\lambda_2 := -1 \quad (3.2)$$

Reduced equation in terms of invariants:

```
> lambda[1]*x^2+lambda[2]*y^2+det(A)/det(A00)=0;
```

$$x^2 - y^2 - 1 = 0 \quad (3.3)$$

The new coordinate system R'={O', B'} has:
- origin O' the center of the conic [1,1].
- basis B' an orthonormal basis given by the eigenvectors of A00.

```
> Q:=matrix(3,3,[1,0,0,1,1,0,1,0,1]);
```

$$Q := \begin{bmatrix} 1 & 0 & 0 \\ 1 & 1 & 0 \\ 1 & 0 & 1 \end{bmatrix} \quad (3.4)$$

```
> Ap:=evalm(transpose(Q)&*A&*Q);Xa:=matrix(3,1,[1,x,y]);
```

$$Ap := \begin{bmatrix} -1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & -1 \end{bmatrix}$$

$$Xa := \begin{bmatrix} 1 \\ x \\ y \end{bmatrix} \quad (3.5)$$

```
> reduced:=simplify(evalm(transpose(Xa)&*Ap&*Xa)[1,1]);
```

$$reduced := x^2 - y^2 - 1 \quad (3.6)$$